/*****Log*****/
clear
cd "C:\Users\Joshua\Dropbox\Work\Networks and Economic Perceptions\Data and Do Files"
	/*the above needs to be updated to where you have the data/do files*/
	
log using "Online Appendix D Output.smcl", replace
	/*the log file is in format .smcl, but can be read via txt editor*/

set more off

/***********************************************************************
************************************************************************
************************************************************************
This do file contains the needed to replicate Online Appendix D where
we investigate alterantive measurements of network disagreement. 

For each survey, the code will load and clean the data at hand using the
commands included in the survey cleaning .do files and then do the 
analyses in question. For those wishing to use this to replicate the analyses, 
you will need to update the cd command below to point to where your data
and .do files are located. 

Also note that there are two STATA modules that need to be intalled: 
(1) esttab and (2) grc1leg. 

ssc install estout, replace
ssc install grc1leg, replace

Note: The analyses here were run using STATA 13. Users of STATA 15 
will likely get an error message when the esstab 
command below runs, specifically: 
	"equation / not found"
This is because of the "rename" function in the command seems to be 
broken in STATA 15. Deleting this line will fix matters. 

************************************************************************
************************************************************************
***********************************************************************/


/****************************************
*****************************************
		1992 CNEP
*****************************************
****************************************/	

*data cleaning*
do "Data Cleaning - 1992 CNEP.do"
set more off


/****Partisan Extremity****/

eststo clear
eststo: ologit pid_str_full c.disagree_total numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
			income i.marital
eststo: ologit pid_str_full c.disagree_avg numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
		income i.marital
			
eststo: ologit pid_str_full c.disagree_total_freq numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
		income i.marital
			
eststo: ologit pid_str_full c.disagree_total_knowl numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
			income i.marital

eststo: ologit pid_str_full c.disagree_total_weight numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
			income i.marital

eststo: ologit pid_str_full c.genavg numgiven disc_knowl i.pid_2 news_att i.gender i.race age educ ///
			income i.marital

			
esttab using 1992_ALT_PIDSTR.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original" "/(D+A)" "Weight: Disc. Freq" "Weight: Disc. Soph." "Weight: Gen Dis." "Gen Dis.") ///
	title({\b Table XX.} "Partisan Extremity Alternative Measures of Disagreement - 1992 CNEP") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement disagree_total_weight Disagreement genavg Disagreement) 
eststo clear
			
			
/****Economic Evaluations****/


eststo clear
eststo: ologit retro i.partisan##c.disagree_total numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 

eststo: ologit retro i.partisan##c.disagree_avg numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 
	
eststo: ologit retro i.partisan##c.disagree_total_freq numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 
	
eststo: ologit retro i.partisan##c.disagree_total_knowl numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 
	
eststo: ologit retro i.partisan##c.disagree_total_weight numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 
	
eststo: ologit retro i.partisan##c.genavg numgiven disc_knowl news_att age educ ///
	income i.gender i.race i.employed i.marital 
	
esttab using 1992_ALT_ECON.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original" "/(D+A)" "Weight: Disc. Freq" "Weight: Disc. Soph." "Weight: Gen Dis." "Gen Dis.") ///
	title({\b Table XX.} "Partisan Extremity Alternative Measures of Disagreement - 1992 CNEP") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement disagree_total_weight Disagreement genavg Disagreement) 
eststo clear
			
/****************************************
*****************************************
		2000 ANES
*****************************************
****************************************/	

clear
do "Data Cleaning - 2000 ANES.do"

/***Knowledge***/
eststo clear
foreach var in disagree_total disagree_avg  disagree_total_freq disagree_total_knowl {
eststo: logit def_knowl i.partisan##c.`var' names1 disc_knowl interest i.gender i.race ///
	i.marital age educ income nfc1 evaluate1 [pweight = V000002a]
	}
			
esttab using 2000_ALTMEASURES_KNOWL.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weight: Disc. Freq" "Weight: Disc Knowl.")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2000 ANES (Knowledge)") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement) 
				
eststo clear


/***Economic Evaluations***/

eststo clear
foreach var in disagree_total disagree_avg  disagree_total_freq disagree_total_knowl {
	eststo: ologit econ_post3 i.partisan##c.`var' names1 disc_knowl interest age educ income ///
	i.gender i.race i.employed i.marital nfc1 evaluate1 [pweight = V000002a]
	}
	
esttab using 2000_ALTMEASURES_ECON.rtf, onecell nobaselevels replace  label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weight: Disc. Freq" "Weight: Disc Knowl.")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2000 ANES (Knowledge)") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement ///
	1.partisan#c.disagree_total Party*Disagreement 1.partisan#c.disagree_avg Party*Disagreement ///
		1.partisan#c.disagree_total_freq Party*Disagreement 1.partisan#c.disagree_total_knowl Party*Disagreement)
				
eststo clear



/****************************************
*****************************************
		2002 ANES
*****************************************
****************************************/				 
		
*data cleaning
clear
do "Data Cleaning - 2002 ANES.do"



/***Partisan Extremity****/
eststo clear
foreach var in disagree_total disagree_avg disagree_total_freq disagree_total_knowl  {
	eststo:  ologit pid_str_full c.`var' names1 disc_knowl i.pid_2_2000 follow  ///
			i.gender i.race age2000 educ2000 income2000  i.marital2000 evaluate1 nfc1 [pweight=WT02PRE]
		}
		

esttab using 2002_ALTMEASURES_EXT.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weigh: Disc. Freq" "Weigh: Disc. Knowl")  ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement) 
eststo clear

/***Economic Evaluations***/
eststo clear
foreach var in disagree_total disagree_avg disagree_total_freq disagree_total_knowl  {
	eststo: ologit retro2002_3 i.partisan2000##c.`var' names1 disc_knowl ///
	follow age2002 educ income2002 i.gender i.race i.employed i.marital nfc1 evaluate1 [pweight=WT02PRE]
	}	
	
esttab using 2002_ALTMEASURES_ECON.rtf, onecell nobaselevels replace  label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weigh: Disc. Freq" "Weigh: Disc. Knowl")  ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_knowl Disagreement ///
		1.partisan2000#c.disagree_total Partisan*Disagreement 1.partisan2000#c.disagree_avg Partisan*Disagreement  ///
		1.partisan2000#c.disagree_total_freq Partisan*Disagreement 1.partisan2000#c.disagree_total_knowl Partisan*Disagreement )
		
eststo clear


/****************************************
*****************************************
		2006 ANES
*****************************************
****************************************/	

*data cleaning
clear
do "Data Cleaning - 2006 ANES.do"
set more off


/***Partisan Extremity***/
eststo clear
foreach var in disagree_total disagree_avg disagree_total_freq disagree_total_int ///
			disagree_total_close disagree_total_weight disc_gen {
	eststo: ologit pid_str_full `var' names disc_interest i.pid_204 follow04   ///
			i.gender i.race age2004 educ income i.marital evaluate1 nfc1 [pweight = V06P002]
	}

esttab using 2006_ALTMEASURES_EXT.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weigh: Disc. Freq" "Weigh: Disc. Interest" ///
		"Weigh: Tie Strength" "Weigh: Gen Dis" "Gen. Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2006 ANES") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement disagree_total_close Disagreement ///
				disagree_total_int Disagreement disc_gen Disagreement disagree_total_weight Disagreement disc_gen Disagreement ///
				1.partisan2004#c.disagree_total Partisan*Disagree ///				
				1.partisan2004#c.disagree_avg Partisan*Disagree ///
				1.partisan2004#c.disagree_total_freq Partisan*Disagree ///
				1.partisan2004#c.disagree_total_int Partisan*Disagree ///
				1.partisan2004#c.disagree_total_close Partisan*Disagree ///
				1.partisan2004#c.disagree_total_weight Partisan*Disagree ///
				1.partisan2004#c.disc_gen Partisan*Disagree )

				
eststo clear


/***Economic Evaluations***/
eststo clear
foreach var in disagree_total disagree_avg disagree_total_freq disagree_total_int ///
			disagree_total_close disagree_total_weight disc_gen {
	eststo: ologit retro i.partisan2004##c.`var' names disc_interest follow04 ///
		age educ income i.gender i.race i.employed i.marital nfc1 evaluate1 [pweight = V06P002]	
	}
	
	
esttab using 2006_ALTMEASURES_ECON.rtf, onecell nobaselevels replace  pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "/(D+A)" "Weigh: Disc. Freq" "Weigh: Disc. Interest" ///
		"Weigh: Tie Strength" "Weigh: Gen Dis" "Gen. Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2006 ANES") ///
	rename(disagree_total Disagreement disagree_avg Disagreement disagree_total_freq Disagreement ///
				disagree_total_int Disagreement disc_gen Disagreement disagree_total_weight Disagreement disc_gen Disagreement) 

eststo clear	


/****************************************
*****************************************
		2008-9 ANES
*****************************************
****************************************/	

clear
do "Data Cleaning - 2008 Panel.do"
set more off

/***Partisan Extremity***/
*W10*
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: ologit pid_str10_full `var' numgiven1 network_interest  i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC10]
	}
	
esttab using 2008OCT_ALTMEASURES_PARTISANSHIP.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (October)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 

eststo clear
	
*W11*
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: ologit pid_str11_full `var' numgiven1 network_interest i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC11]
		}	

esttab using 2008NOV_ALTMEASURES_PARTISANSHIP.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (November)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 

eststo clear

*W17*
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
eststo: ologit pid_str17_full `var' numgiven1 network_interest i.pid_21 interest_w1 ///
			i.gender i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC17]
			}
			
esttab using 2008MAY_ALTMEASURES_PARTISANSHIP.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (May 2009)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 

eststo clear
			
*W19
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
eststo: ologit pid_str19_full `var' numgiven1 network_interest i.pid_21 interest_w1 i.gender ////
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC19]
	}	

esttab using 2008JULY_ALTMEASURES_PARTISANSHIP.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (July 2009)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 

eststo clear

	
	
/***Partisan Ambivalence****/
*in likes
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: ologit in_like_w11  `var'  numgiven1 network_interest  i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC11]
	}		
	
esttab using 2008AMBIV_ALTMEASURES_INLIKE.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (AMBIVALENCE: IN-LIKES)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 


*out dislike

eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
eststo: ologit out_dislike_w11 `var'  numgiven1 network_interest  i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC11]
	}
	
esttab using 2008AMBIV_ALTMEASURES_OUTDISLIKE.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (AMBIVALENCE: OUT-DISLIKE)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 


*in dislike	
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {	
eststo: ologit in_dislike_w11 `var'  numgiven1 network_interest  i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC11]
}


esttab using 2008AMBIV_ALTMEASURES_INDISLIKE.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (AMBIVALENCE: IN-DISLIKE)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 


*out like			

eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
eststo: ologit out_like_w11 `var'  numgiven1 network_interest  i.pid_21 interest_w1 i.gender ///
			i.race age educ income i.marital evaluate1 nfc1 [pweight= WGTC11]
	}


esttab using 2008AMBIV_ALTMEASURES_OUTLIKE.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (AMBIVALENCE: OUT-LIKE)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement) 

/***Knowledge****/
*W17*
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: logit def_knowl17 i.partisan_9rev##c.`var' numgiven1 network_interest interest_w9 i.gender i.race i.marital ///
	age educ income nfc1 evaluate1  [pweight= WGTPP17]
}

esttab using 2008KNOWL_ALTMEASURES_MAY.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (KNOWLEDGE: MAY)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement ///
				1.partisan_9rev#c.disagree_total Partisan*Disagree ///
				1.partisan_9rev#c.disagree_avg Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_int Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_close Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_weight Partisan*Disagree ///
				1.partisan_9rev#c.gendiff Partisan*Disagree)

*W19
eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: logit def_knowl19 i.partisan_9rev##c.`var' ///
	numgiven1 network_interest interest_w9 i.gender i.race i.marital age educ income nfc1 evaluate1 [pweight= WGTPP19]
		}
		
		
esttab using 2008KNOWL_ALTMEASURES_JULY.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (KNOWLEDGE: JULY)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement ///
				1.partisan_9rev#c.disagree_total Partisan*Disagree ///
				1.partisan_9rev#c.disagree_avg Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_int Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_close Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_weight Partisan*Disagree ///
				1.partisan_9rev#c.gendiff Partisan*Disagree)

/***Economic Evaluations***/

*W17


eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: ologit w17_retro i.partisan_9rev##c.`var' c.numgiven network_interest interest_w9 	///
	age educ income i.gender i.race i.empl_w11a i.marital nfc1 evaluate1 [pweight= WGTPP17]
	
	}
	
esttab using 2008ECON_ALTMEASURES_MAY.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (ECON MAY)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement ///
				1.partisan_9rev#c.disagree_total Partisan*Disagree ///
				1.partisan_9rev#c.disagree_avg Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_int Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_close Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_weight Partisan*Disagree ///
				1.partisan_9rev#c.gendiff Partisan*Disagree)
	
	
*W19

eststo clear
foreach var in disagree_total disagree_avg disagree_total_int disagree_total_close  ///
	disagree_total_weight gendiff  {
	eststo: ologit w19_retro i.partisan_9rev##c.`var' c.numgiven network_interest interest_w9 ///
		age educ income i.gender i.race i.empl_w11a i.marital nfc1 evaluate1 [pweight= WGTPP19]
	}

esttab using 2008ECON_ALTMEASURES_JULY.rtf, onecell nobaselevels replace label pr2 aic bic se star(+ 0.10 * 0.05 ** 0.01) ///
	mtitles("Original Measure" "(/D+A)" "Weigh: by Disc. Interest" ///
		"Weigh: by Tie Strength" "Weighted by Gen Dis" "Gen Disagree")  ///
	title({\b Table XX.} "Alternative Measures of Disagreement - 2008-2009 ANES (Econ: July)") ///
	rename(disagree_total Disagreement disagree_avg "Disagreement" disagree_total_int Disagreement disagree_total_close Disagreement ///
				disagree_total_weight Disagreement gendiff Disagreement ///
				1.partisan_9rev#c.disagree_total Partisan*Disagree ///
				1.partisan_9rev#c.disagree_avg Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_int Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_close Partisan*Disagree ///
				1.partisan_9rev#c.disagree_total_weight Partisan*Disagree ///
				1.partisan_9rev#c.gendiff Partisan*Disagree)

				
eststo clear




log close





		